﻿<%
 'OPTION EXPLICIT
 const BASE_64_MAP_INIT = "ABCDEFGHIJKLMNOPQRSTUVWXYZabcdefghijklmnopqrstuvwxyz0123456789+/"
 Dim newline
 Dim Base64EncMap(63)
 Dim Base64DecMap(127) 
 '初始化函数
 PUBLIC SUB initCodecs()
	  ' 初始化变量
	  newline = "<P>" & chr(13) & chr(10)
	  Dim max, idx
		 max = len(BASE_64_MAP_INIT)
	  For idx = 0 to max - 1
		   Base64EncMap(idx) = mid(BASE_64_MAP_INIT, idx + 1, 1)
	  Next
	  For idx = 0 to max - 1
		   Base64DecMap(ASC(Base64EncMap(idx))) = idx
	  Next
 END SUB
 'Base64加密函数
 PUBLIC FUNCTION base64Encode(plain)
	  If len(plain) = 0 Then
		   base64Encode = ""
		   exit function
	  End If
	  Dim ret, ndx, by3, first, second, third
	  by3 = (len(plain) \ 3) * 3
	  ndx = 1
	  do while ndx <= by3
		   first  = asc(mid(plain, ndx+0, 1))
		   second = asc(mid(plain, ndx+1, 1))
		   third  = asc(mid(plain, ndx+2, 1))
		   ret = ret & Base64EncMap(  (first \ 4) AND 63 )
		   ret = ret & Base64EncMap( ((first * 16) AND 48) + ((second \ 16) AND 15 ) )
		   ret = ret & Base64EncMap( ((second * 4) AND 60) + ((third \ 64) AND 3 ) )
		   ret = ret & Base64EncMap( third AND 63)
		   ndx = ndx + 3
	  loop
	  If by3 < len(plain) Then
		   first  = asc(mid(plain, ndx+0, 1))
		   ret = ret & Base64EncMap(  (first \ 4) AND 63 )
		   If (len(plain) MOD 3 ) = 2 Then
				second = asc(mid(plain, ndx+1, 1))
				ret = ret & Base64EncMap( ((first * 16) AND 48) + ((second \ 16) AND 15 ) )
				ret = ret & Base64EncMap( ((second * 4) AND 60) )
		   Else
				ret = ret & Base64EncMap( (first * 16) AND 48)
				ret = ret '& "="
		   End If
		   ret = ret '& "="
	  End If
	  base64Encode = ret
 END FUNCTION
 'Base64解密函数
 PUBLIC FUNCTION base64Decode(scrambled)
	  If len(scrambled) = 0 Then
		   base64Decode = ""
		   exit function
	  End If
	  Dim realLen
	  realLen = len(scrambled)
	  do while mid(scrambled, realLen, 1) = "="
		   realLen = realLen - 1
	  loop
	  Dim ret, ndx, by4, first, second, third, fourth
	  ret = ""
	  by4 = (realLen \ 4) * 4
	  ndx = 1
	  do while ndx <= by4
		   first  = Base64DecMap(asc(mid(scrambled, ndx+0, 1)))
		   second = Base64DecMap(asc(mid(scrambled, ndx+1, 1)))
		   third  = Base64DecMap(asc(mid(scrambled, ndx+2, 1)))
		   fourth = Base64DecMap(asc(mid(scrambled, ndx+3, 1)))
		   ret = ret & chr( ((first * 4) AND 255) +   ((second \ 16) AND 3))
		   ret = ret & chr( ((second * 16) AND 255) + ((third \ 4) AND 15))
		   ret = ret & chr( ((third * 64) AND 255) +  (fourth AND 63))
		   ndx = ndx + 4
	  loop
	  If ndx < realLen Then
		   first  = Base64DecMap(asc(mid(scrambled, ndx+0, 1)))
		   second = Base64DecMap(asc(mid(scrambled, ndx+1, 1)))
		   ret = ret & chr( ((first * 4) AND 255) +   ((second \ 16) AND 3))
		   If realLen MOD 4 = 3 Then
				third = Base64DecMap(asc(mid(scrambled,ndx+2,1)))
				ret = ret & chr( ((second * 16) AND 255) + ((third \ 4) AND 15))
		   End If
	  End If
	  base64Decode = ret
 END FUNCTION
' 初始化
 call initCodecs                   
' 测试代码

%>